

SUBPROCESOS

Para definir un subproceso, la sintaxis sería más o menos así:

```
SubProceso valor de retorno = NombreSubProc (arg1, arg2,  
arg3)  
    // ...hacer algo con los argumentos y...  
    // ...guardar algo en la variable valor de retorno...  
FinSubProceso
```

Luego, se podría invocar el subproceso como una función más, en cualquier expresión, o utilizar como una instrucción más. Un ejemplo concreto:

```
SubProceso R <- Factorial(x)  
    R <- 1; i <- x;  
    Mientras i>1 Hacer  
        R <- R*i;  
        i <- i-1;  
    FinMientras  
FinSubProceso
```

```
SubProceso x <- LeerDato(cosa)  
    Escribir "Ingrese ", cosa, ": ";  
    Leer x;  
FinSubProceso
```

```
SubProceso ImprimirResultado(x)  
    Escribir "El resultado es: ", x  
FinSubProceso
```

```
Proceso UsaFunciones  
    dato <- LeerDato("un número entero mayor a 1");  
    resultado <- Factorial(dato);  
    ImprimirResultado(resultado);  
FinProceso
```

El ejemplo declara 3 subprocesos. 2 de ellos devuelven cosas y por eso se usan como parte de expresiones, el último no devuelve nada, se lo invoca como a una instrucción más. Para quien conozca el manejo de funciones en otros lenguajes, debo decir que los argumentos se pasan siempre por copia (no hay pasaje por referencia) y que el subproceso retorna un solo valor o ninguno (no puede retornar más de una variable)

SUBPROCESOS

```
// funcion que no recibe ni devuelve nada  
SubProceso Saludar  
    Escribir "Hola mundo!"  
FinSubProceso
```

```
// funcion que recibe un argumento por valor, y  
devuelve su doble
```

```

SubProceso res <- CalcularDoble(num)
    res <- num*2 // retorna el doble
FinSubProceso

// funcion que recibe un argumento por referencia,
y lo modifica
SubProceso Triplicar(num por referencia)
    num <- num*3 // modifica la variable
duplicando su valor
FinSubProceso

// proceso principal, que invoca a las funciones
antes declaradas

Proceso PruebaFunciones

    Escribir "Llamada a la funcion Saludar:"
    Saludar // como no recibe argumentos pueden
omitirse los paréntesis vacios

    Escribir "Ingrese un valor numérico para x:"
    Leer x

    Escribir "Llamada a la función CalcularDoble
(pasaje por valor)"
    Escribir "El doble de ",x," es ",
CalcularDoble(x)
    Escribir "El valor original de x es ",x

    Escribir "Llamada a la función Triplicar
(pasaje por referencia)"
    Triplicar(x)
    Escribir "El nuevo valor de x es ", x

FinProceso

```